```
> restart;
```

Groebner bases.

Michael Monagan

```
> with(Groebner):
```

Compute a Groebner basis for the ideal generated by $f_1, f_2$ below.

```
> f := [x^2+y^2-1,x*y-1];
```

$$f := [x^2 + y^2 - 1, x\,y - 1] \quad (1)$$

```
> G1 := f;
```

$$G1 := [x^2 + y^2 - 1, x\,y - 1] \quad (2)$$

```
> s := SPolynomial(f[1],f[2],plex(x,y));
```

$$s := y^3 + x - y \quad (3)$$

```
> r := NormalForm(s,G1,plex(x,y));
```

$$r := y^3 + x - y \quad (4)$$

```
> f := [op(f),r];
```

$$f := [x^2 + y^2 - 1, x\,y - 1, y^3 + x - y] \quad (5)$$

```
> G2 := f;
```

$$G2 := [x^2 + y^2 - 1, x\,y - 1, y^3 + x - y] \quad (6)$$

```
> for i to nops(f) do
      for j from i+1 to nops(f) do
          r := NormalForm( SPolynomial(f[i],f[j],plex(x,y)), G2, plex
  (x,y) );
          printf("S(f[%d],f[%d]) mod G2 = %a\n",i,j,r);
      od;
  od;
S(f[1],f[2]) mod G2 = 0
S(f[1],f[3]) mod G2 = 0
S(f[2],f[3]) mod G2 = -y^4+y^2-1
```

```
> f4 := -y^4+y^2-1;
```

$$f4 := -y^4 + y^2 - 1 \quad (7)$$

```
> f := [op(f),f4];
  G3 := f;
```

$$f := [x^2 + y^2 - 1, x\,y - 1, y^3 + x - y, -y^4 + y^2 - 1]$$

$$G3 := [x^2 + y^2 - 1, x\,y - 1, y^3 + x - y, -y^4 + y^2 - 1] \quad (8)$$

```
> for i to nops(f) do
      for j from i+1 to nops(f) do
          r := NormalForm( SPolynomial(f[i],f[j],plex(x,y)), G3, plex
  (x,y) );
          printf("S(f[%d],f[%d]) mod G3 = %a\n",i,j,r);
      od;
  od;
S(f[1],f[2]) mod G3 = 0
S(f[1],f[3]) mod G3 = 0
S(f[1],f[4]) mod G3 = 0
S(f[2],f[3]) mod G3 = 0
S(f[2],f[4]) mod G3 = 0
S(f[3],f[4]) mod G3 = 0
```

```
> G3;
```

(9)

$$[x^2 + y^2 - 1, x\,y - 1, y^3 + x - y, -y^4 + y^2 - 1] \quad (9)$$

To make G3 a reduced Groebner basis

```
> for i to nops(G3) do G3[i] := NormalForm(G3[i],subsop(i=NULL,G3),
  plex(x,y)); od:
  G3;
```

$$[0, 0, y^3 + x - y, -y^4 + y^2 - 1] \quad (10)$$

```
> G3 := [-G3[4],G3[3]];
```

$$G3 := [y^4 - y^2 + 1, y^3 + x - y] \quad (11)$$

```
> Groebner[Basis]( [f[1],f[2]], plex(x,y) );
```

$$[y^4 - y^2 + 1, y^3 + x - y] \quad (12)$$

Example of solving a polynomial system

```
> f := [x^2+y^2-2,x*y-1];
```

$$f := [x^2 + y^2 - 2, x\,y - 1] \quad (13)$$

```
> G := Groebner[Basis]( f, plex(x,y) );
```

$$G := [y^4 - 2\,y^2 + 1, y^3 + x - 2\,y] \quad (14)$$

A Groenber basis for $\langle f_1, f_2 \rangle \cap \mathbb{Q}[y]$ is $\{G_1 = y^4 - 2 \cdot y^2 + 1\}$.

```
> g := G[1];
```

$$g := y^4 - 2\,y^2 + 1 \quad (15)$$

```
> solve(g=0,y);
```

$$1, 1, -1, -1 \quad (16)$$

```
> eval(G,y=1);
```

$$[0, x - 1] \quad (17)$$

```
> solve(eval(G,y=1));
```

$$\{x = 1\} \quad (18)$$

```
> eval(G,y=-1);
```

$$[0, x + 1] \quad (19)$$

```
> solve(eval(G,y=-1),x);
```

$$\{x = -1\} \quad (20)$$

Thus the solutions are $\{x = 1, y = 1\}, \{x = -1, y = -1\}$.

```
> solve(f);
```

$$\{x = 1, y = 1\}, \{x = -1, y = -1\} \quad (21)$$